Pensieve header: A unified verification notebook for the PPSA project.

Continues pensieve://2017-06/ and pensieve://2017-08/.

Prolog

```
Go;
```

```
SetOptions[EvaluationNotebook[],
  NotebookEventActions → {
    {"MenuCommand", "Save"} ⧴ (
      SetDirectory["C:\\drorbn\\AcademicPensieve\\Projects\\PPSA"];
      Get["MakeVSnips.m"]
     ),
    PassEventsDown → True
   }];
```

```
HL[ℰ_] := Style[ℰ, Background → Yellow];
```

Initialization

The “degree carrier / filtration parameter” is $\hbar$, and all “coupling constants” are proportional to it.

TD

```
$TℏD = 5; ℏ /: ℏ^d_. /; d > $TℏD := 0;
$TϵD = 2; ϵ /: ϵ^d_. /; d > $TϵD := 0;
SetAttributes[SS, HoldAll];
SS[ℰ_] := Block[{ℏ, ϵ},
  Normal@Series[ℰ, {ℏ, 0, $TℏD}, {ϵ, 0, $TϵD}] ]
Simp[ℰ_] := Collect[ℰ, _CU | _QU, Expand];
```

DeclareAlgebra

QLImplementation

```
Unprotect[NonCommutativeMultiply];
NonCommutativeMultiply[x_] := x; 0 ** _ = _ ** 0 = 0;
B[x_, x_] = 0; B[x_, y_] := x ** y - y ** x;
(x_Plus) ** y_ := (# ** y) & /@ x; x_ ** (y_Plus) := (x ** #) & /@ y;
```

QLImplementation

```
DeclareAlgebra[U_Symbol, opts__Rule] := Module[{gp, sr, cp, S, pow,
    gs = Generators /. {opts}, cs = Centrals /. {opts}
  },
  gp = Alternatives @@ gs;
  gp = gp | gp_; (* generators pattern *)
  sr = Thread[gs → Range@Length@gs]; (* sorting rule *)
  cp = Alternatives @@ cs; (* centrals pattern *)
  S[ℰ_] := Collect[ℰ, _U, Expand];
  U_i_[ℰ_] := ℰ /. {t : cp :→ t_i, u_U :→ Replace[u, x_ :→ x_i, 1]};
  B[U@(x_)_i_, U@(y_)_i_] := B[U@x_i, U@y_i] = U_i@B[U@x, U@y];
  B[U@(x_)_i_, U@(y_)_j_] /; i =!= j := 0;
  B[U@y_, U@x_] := S[-B[U@x, U@y]];
  x_ ** U[] := x; U[] ** x_ := x;
  (a_ * x_U) ** (b_ * y_U) := If[a b === 0, 0, S[a b (x ** y)]];
  (a_ * x_U) ** y_ := S[a (x ** y)]; x_ ** (a_ * y_U) := S[a (x ** y)];
  U[xx___, x_] ** U[y_, yy___] := If[OrderedQ[{x, y} /. sr],
    U[xx, x, y, yy], U@xx ** (U@y ** U@x + B[U@x, U@y]) ** U@yy];
  U@{c_. * (l : gp)^n_, r___} /; FreeQ[c, gp] := S[c U@Table[l, {n}] ** U@{r}];
  U@{c_. * l : gp, r___} := S[c U[l] ** U@{r}];
  U@{c_, r___} /; FreeQ[c, gp] := S[c U@{r}];
  U@{} = U[];
  U@{l_Plus, r___} := S[U@{#, r} & /@ l];
  U@{l_, r___} := U@{Expand[l], r};
  O_U[poly_, specs___] := Module[{sp, null, vs, us},
    sp = Replace[{specs}, l_List :→ (l → null), {1}];
    vs = Join @@ (First /@ sp);
    us = Join @@ (sp /. (l_ → s_) :→ (l /. x_i_ :→ x_s));
    S[Total[
        CoefficientRules[poly, vs] /. (p_ → c_) :→ c U@(us^p)
      ]] /. x_null :→ x
    ];
  pow[ℰ_, 0] = U[]; pow[ℰ_, n_] := pow[ℰ, n - 1] ** ℰ;
  𝕊_U[ℰ_, ss___Rule] := S@Total[
     CoefficientRules[ℰ, First /@ {ss}] /.
      (p_ → c_) :→ c NonCommutativeMultiply @@ MapThread[pow, {Last /@ {ss}, p}]]
  ]
```

DeclareMorphism

QLImplementation

```
DeclareMorphism[m_, U_ → V_, ongs_List, oncs_List: {}] := (
  Replace[ongs, (g_ → img_) :→ (m[U[g]] = img), {1}];
  m[U[]] = V[];
  m[U[g_i_]] := V_i[m[U@g]];
  m[U[vs__]] := NonCommutativeMultiply @@ (m /@ U /@ {vs});
  m[ℰ_] := Simp[ℰ /. oncs /. u_U :→ m[u]]; )
```

Implementing $\mathrm{sl}_2^{\gamma\epsilon}$

CU

```
DeclareAlgebra[CU, Generators → {y, a, x}, Centrals → {t}];
B[CU@a, CU@y] = -γ CU@y; B[CU@x, CU@a] = -γ CU@x;
B[CU@x, CU@y] = 2 ∈ CU@a - t CU[];
```

Verifying associativity on triples of generators:

```
With[{bas = CU /@ {y, a, x}},
 Table[z1 ** (z2 ** z3) - (z1 ** z2) ** z3 // Simp // HL,
  {z1, bas}, {z2, bas}, {z3, bas} ] ]
```

```
{{{0, 0, 0}, {0, 0, 0}, {0, 0, 0}},
 {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}, {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}}
```

Verifying associativity on a “random” triple:

```
With[{z1 = CU[y, y, a, a, x, x], z2 = CU[y, a, x], z3 = CU[y, y, a, x]}, {
   (rhs = (z1 ** z2) ** z3 // Simp) // Short,
   z1 ** (z2 ** z3) - rhs // Simp // HL
  }] // Timing
```

```
{0.5625, {(28 t^2 γ^4 + 116 t γ^5 ∈ + 120 γ^6 ∈^2) CU[y, y, y, x, x] +
   <<23>> + CU[y, y, y, y, y, a, a, a, a, x, x, x, x], 0}}
```

Implementing $\mathcal{U}_{\gamma\epsilon;\hbar}$

With $q = e^{\hbar\gamma\epsilon}$, $A = e^{-\hbar\epsilon a}$, $T = e^{\hbar t/2}$, and $[f, g]_q := fg - qgf$, our algebra is $\mathcal{U}_{\gamma\epsilon;\hbar} = \langle t, y, a, x\rangle/\mathcal{R}$, where $\mathcal{R} = \left([t, *] = 0,\ [a, y] = -\gamma y,\ [x, y]_q = \left(1 - T^2 A^2\right)/\hbar,\ [x, a] = -\gamma x\right)$.

QU

```
DeclareAlgebra[QU, Generators → {y, a, x}, Centrals → {t, T}];
q := SS[e^(γ ∈ ℏ)]; T := SS[e^(ℏ t/2)];
B[QU@a, QU@y] = -γ QU@y; B[QU@x, QU@a] = -γ QU@x;
B[QU@x, QU@y] := (q - 1) QU@{y, x} + 𝕆_QU[SS[(1 - e^(ℏ (t-2 ∈ a))) / ℏ], {a}];
```

```
With[{bas = QU /@ {y, a, x}},
 Table[{z1, z2} -> z1 ** z2 - z2 ** z1 // Simp, {z1, bas}, {z2, bas}] ]
```

$$\{\{\{QU[y], QU[y]\} \to 0, \{QU[y], QU[a]\} \to \gamma\, QU[y],$$
$$\{QU[y], QU[x]\} \to \left(t + \frac{t^2 \hbar}{2} + \frac{t^3 \hbar^2}{6} + \frac{t^4 \hbar^3}{24} + \frac{t^5 \hbar^4}{120} + \frac{t^6 \hbar^5}{720}\right) QU[\,] +$$
$$\left(-2 \epsilon - 2 t \epsilon \hbar - t^2 \epsilon \hbar^2 - \frac{1}{3} t^3 \epsilon \hbar^3 - \frac{1}{12} t^4 \epsilon \hbar^4 - \frac{1}{60} t^5 \epsilon \hbar^5\right) QU[a] +$$
$$\left(2 \epsilon^2 \hbar + 2 t \epsilon^2 \hbar^2 + t^2 \epsilon^2 \hbar^3 + \frac{1}{3} t^3 \epsilon^2 \hbar^4 + \frac{1}{12} t^4 \epsilon^2 \hbar^5\right) QU[a, a] + \left(-\gamma \epsilon \hbar - \frac{1}{2} \gamma^2 \epsilon^2 \hbar^2\right) QU[y, x]\},$$
$$\{\{QU[a], QU[y]\} \to -\gamma\, QU[y], \{QU[a], QU[a]\} \to 0, \{QU[a], QU[x]\} \to \gamma\, QU[x]\},$$
$$\{\{QU[x], QU[y]\} \to \left(-t - \frac{t^2 \hbar}{2} - \frac{t^3 \hbar^2}{6} - \frac{t^4 \hbar^3}{24} - \frac{t^5 \hbar^4}{120} - \frac{t^6 \hbar^5}{720}\right) QU[\,] +$$
$$\left(2 \epsilon + 2 t \epsilon \hbar + t^2 \epsilon \hbar^2 + \frac{1}{3} t^3 \epsilon \hbar^3 + \frac{1}{12} t^4 \epsilon \hbar^4 + \frac{1}{60} t^5 \epsilon \hbar^5\right) QU[a] +$$
$$\left(-2 \epsilon^2 \hbar - 2 t \epsilon^2 \hbar^2 - t^2 \epsilon^2 \hbar^3 - \frac{1}{3} t^3 \epsilon^2 \hbar^4 - \frac{1}{12} t^4 \epsilon^2 \hbar^5\right) QU[a, a] + \left(\gamma \epsilon \hbar + \frac{1}{2} \gamma^2 \epsilon^2 \hbar^2\right) QU[y, x],$$
$$\{QU[x], QU[a]\} \to -\gamma\, QU[x], \{QU[x], QU[x]\} \to 0\}\}$$

Verifying associativity on triples of generators:

```
With[{bas = QU /@ {y, a, x}},
 Table[HL[z1 ** (z2 ** z3) - (z1 ** z2) ** z3] // Simp,
  {z1, bas}, {z2, bas}, {z3, bas} ] ]
```

{{{0, 0, 0}, {0, 0, 0}, {0, 0, 0}},
 {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}, {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}}

Verifying associativity on a “random” triple (~34 secs @ T\hbar$D=5, T\epsilon$D=2):

```
With[{z1 = QU[y, y, a, a, x, x], z2 = QU[y, a, x], z3 = QU[y, y, a, x]}, {
   (rhs = (z1 ** z2) ** z3 // Simp) // Short,
   HL[z1 ** (z2 ** z3) - rhs // Simp ]
  }] // Timing
```

$$\left\{112.234, \left\{\left(28 t^2 \gamma^4 + 116 t \gamma^5 \epsilon + 120 \gamma^6 \epsilon^2 + \ll 12 \gg + \frac{7}{10} t^7 \gamma^4 \hbar^5 + \frac{1549}{90} t^6 \gamma^5 \epsilon \hbar^5 + \frac{2215}{12} t^5 \gamma^6 \epsilon^2 \hbar^5\right) QU[y, \ll 3 \gg, x] + \ll 22 \gg, 0\right\}\right\}$$

Verifying that $\mathrm{Lim}_{\hbar \to 0}$ QU = CU using a “random” product (~23 secs @ T\hbar$D=5, T\epsilon$D=2):

```
With[{z1 = CU[y, y, a, a, x, x], z2 = CU[y, a, x], z3 = CU[y, y, a, x]}, {
   Short[lhs = z1 ** (z2 ** z3)],
   Short[rhs = (QU @@ z1) ** ((QU @@ z2) ** (QU @@ z3))],
   Expand[Limit[rhs /. {QU → CU, T → ⅇ^(ℏ t/2)}, ℏ → 0] == lhs] // HL
  }] // Timing
```

$$\{37.4063, \{2 (8 t^2 \gamma^4 + 16 t \gamma^5 \epsilon)\, CU[y, y, y, x, x] + \ll 94 \gg + CU[y, y, y, y, y, a, a, a, a, x, x, x, x], \ll 1 \gg, \mathrm{True}\}\}$$

Implementing θ

theta

```
DeclareMorphism[Cθ, CU → CU, {y → -CU@x, a → -CU@a, x → -CU@y}, {t → -t, T → T^-1}];
DeclareMorphism[Qθ, QU → QU, {y → 𝕆_QU[SS[-T^-1 ⅇ^(ℏ ϵ a) x], {a, x}],
  a → -QU@a, x → 𝕆_QU[SS[-T^-1 ⅇ^(ℏ ϵ a) y], {a, y}]}, {t → -t, T → T^-1}]
```

Verifying involutivity on CU:

```
With[{bas = CU /@ {y, a, x}},
 Table[z → Cθ[z] → HL[Cθ[Cθ[z]]], {z, bas}] ]
```

{CU[y] → -CU[x] → CU[y], CU[a] → -CU[a] → CU[a], CU[x] → -CU[y] → CU[x]}

Verifying that θ is a multiplicative homomorphism on CU:

```
With[{bas = CU /@ {y, a, x}},
 Table[Cθ[z1 ** z2] - Cθ[z1] ** Cθ[z2] // HL, {z1, bas}, {z2, bas}] ]
```

{{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}

Verifying involutivity on QU:

```
With[{bas = QU /@ {y, a, x}},
 Table[z → Qθ[z] → HL[Qθ[Qθ[z]]], {z, bas}] ]
```

$$\Big\{\mathrm{QU}[y]\to\left(-1+\frac{t\hbar}{2}-\frac{t^2\hbar^2}{8}+\frac{t^3\hbar^3}{48}-\frac{t^4\hbar^4}{384}+\frac{t^5\hbar^5}{3840}\right)\mathrm{QU}[x]+\left(-\epsilon\hbar+\frac{1}{2}t\epsilon\hbar^2-\frac{1}{8}t^2\epsilon\hbar^3+\frac{1}{48}t^3\epsilon\hbar^4-\frac{1}{384}t^4\epsilon\hbar^5\right)\mathrm{QU}[a,x]+\left(-\frac{1}{2}\epsilon^2\hbar^2+\frac{1}{4}t\epsilon^2\hbar^3-\frac{1}{16}t^2\epsilon^2\hbar^4+\frac{1}{96}t^3\epsilon^2\hbar^5\right)\mathrm{QU}[a,a,x]\to\mathrm{QU}[y],\ \mathrm{QU}[a]\to-\mathrm{QU}[a]\to\mathrm{QU}[a],$$

$$\mathrm{QU}[x]\to\left(-1+\frac{t\hbar}{2}+\gamma\epsilon\hbar-\frac{t^2\hbar^2}{8}-\frac{1}{2}t\gamma\epsilon\hbar^2-\frac{1}{2}\gamma^2\epsilon^2\hbar^2+\frac{t^3\hbar^3}{48}+\frac{1}{8}t^2\gamma\epsilon\hbar^3+\frac{1}{4}t\gamma^2\epsilon^2\hbar^3-\frac{t^4\hbar^4}{384}-\frac{1}{48}t^3\gamma\epsilon\hbar^4-\frac{1}{16}t^2\gamma^2\epsilon^2\hbar^4+\frac{t^5\hbar^5}{3840}+\frac{1}{384}t^4\gamma\epsilon\hbar^5+\frac{1}{96}t^3\gamma^2\epsilon^2\hbar^5\right)\mathrm{QU}[y]+\left(-\epsilon\hbar+\frac{1}{2}t\epsilon\hbar^2+\gamma\epsilon^2\hbar^2-\frac{1}{8}t^2\epsilon\hbar^3-\frac{1}{2}t\gamma\epsilon^2\hbar^3+\frac{1}{48}t^3\epsilon\hbar^4+\frac{1}{8}t^2\gamma\epsilon^2\hbar^4-\frac{1}{384}t^4\epsilon\hbar^5-\frac{1}{48}t^3\gamma\epsilon^2\hbar^5\right)\mathrm{QU}[y,a]+\left(-\frac{1}{2}\epsilon^2\hbar^2+\frac{1}{4}t\epsilon^2\hbar^3-\frac{1}{16}t^2\epsilon^2\hbar^4+\frac{1}{96}t^3\epsilon^2\hbar^5\right)\mathrm{QU}[y,a,a]\to\mathrm{QU}[x]\Big\}$$

Verifying that θ is a multiplicative homomorphism on QU:

```
With[{bas = QU /@ {y, a, x}},
 Table[{z1, z2} → HL[Simp[Qθ[z1 ** z2] - Qθ[z1] ** Qθ[z2]]], {z1, bas}, {z2, bas}] ]
```

{{{QU[y], QU[y]} → 0, {QU[y], QU[a]} → 0, {QU[y], QU[x]} → 0},
 {{QU[a], QU[y]} → 0, {QU[a], QU[a]} → 0, {QU[a], QU[x]} → 0},
 {{QU[x], QU[y]} → 0, {QU[x], QU[a]} → 0, {QU[x], QU[x]} → 0}}

The Asymmetric Dequantizator

Following pensieve://People/VanDerVeen/Dequant1.pdf.

ADeq

$$\mathbb{AD}\$f = \frac{\gamma}{\hbar} e^{\hbar\left(\frac{t}{2}-(a+\gamma)\epsilon\right)} \left(\left(\mathrm{Cosh}\left[\hbar\left(a\,\epsilon + \frac{\gamma\,\epsilon}{2} - \frac{t}{2}\right)\right] - \mathrm{Cosh}\left[\hbar\sqrt{\left(\frac{t-\gamma\,\epsilon}{2}\right)^2 + \epsilon\,\omega}\,\right]\right) \Big/ \left(\mathrm{Sinh}\left[\frac{\gamma\,\epsilon\,\hbar}{2}\right]\left(a^2\,\epsilon + a\,\gamma\,\epsilon - a\,t - \omega\right)\right)\right);$$

```
SS[AD$f]
```

$$1 + \left(\frac{t}{2} + (-a-\gamma)\,\epsilon\right)\hbar + \left(\frac{t^2}{6} + \left(\frac{7a^2}{12} + \frac{13a\gamma}{12} + \frac{\gamma^2}{2}\right)\epsilon^2 + \epsilon\left(-\frac{7at}{12} - \frac{7t\gamma}{12} + \frac{\omega}{12}\right)\right)\hbar^2 +$$
$$\left(\frac{t^3}{24} + \epsilon\left(-\frac{5at^2}{24} - \frac{5t^2\gamma}{24} + \frac{t\omega}{24}\right) + \epsilon^2\left(\frac{3a^2t}{8} + \frac{17at\gamma}{24} + \frac{t\gamma^2}{3} - \frac{a\omega}{12} - \frac{\gamma\omega}{12}\right)\right)\hbar^3 +$$
$$\left(\frac{t^4}{120} + \epsilon\left(-\frac{13at^3}{240} - \frac{13t^3\gamma}{240} + \frac{t^2\omega}{80}\right) + \epsilon^2\left(\frac{101a^2t^2}{720} + \frac{4}{15}at^2\gamma + \frac{91t^2\gamma^2}{720} - \frac{2at\omega}{45} - \frac{11t\gamma\omega}{240} + \frac{\omega^2}{360}\right)\right)\hbar^4 +$$
$$\left(\frac{t^5}{720} + \epsilon\left(-\frac{at^4}{90} - \frac{t^4\gamma}{90} + \frac{t^3\omega}{360}\right) + \epsilon^2\left(\frac{3a^2t^3}{80} + \frac{103at^3\gamma}{1440} + \frac{49t^3\gamma^2}{1440} - \frac{1}{72}at^2\omega - \frac{7}{480}t^2\gamma\omega + \frac{t\omega^2}{720}\right)\right)\hbar^5$$

Scaling behaviour of AD$f:

```
Simplify[AD$f == ((AD$f /. γ → 1) /. {ϵ → γ ϵ, a → γ^-1 a, ω → γ^-1 ω})]
```

True

```
FullSimplify[
 AD$f == ((AD$f /. γ → 1) /. {ħ → γ^2 ħ, ϵ → ϵ / γ, a → a / γ, t → γ^-2 t, ω → γ^-3 ω})]
```

True

ADeq

```
AD$ω = γ CU[y, x] + ϵ CU[a, a] - (t - γ ϵ) CU[a];
```

ADeq

```
DeclareMorphism[AD, QU → CU,
  {a → CU@a, x → CU@x, y → S_CU[SS[AD$f], a → CU[a], ω → AD$ω] ** CU@y}]
```

Verifying that the asymmetric dequantizator is a homomorphism:

```
With[{bas = QU /@ {y, a, x}},
 Table[{z1, z2} → HL[Simp[AD[z1 ** z2] - AD[z1] ** AD[z2]]], {z1, bas}, {z2, bas}] ]
```

```
{{{QU[y], QU[y]} → 0, {QU[y], QU[a]} → 0, {QU[y], QU[x]} → 0},
 {{QU[a], QU[y]} → 0, {QU[a], QU[a]} → 0, {QU[a], QU[x]} → 0},
 {{QU[x], QU[y]} → 0, {QU[x], QU[a]} → 0, {QU[x], QU[x]} → 0}}
```

The Symmetric Dequantizator

Following pensieve://People/VanDerVeen/Dequant1.pdf.

SDeq

```
S𝔻$g = √((Cosh[ℏ/2 √(t^2 + γ^2 ϵ^2 + 4 ϵ ϖ)] - Cosh[ℏ/2 (t - (2 a + γ) ϵ)]) /
    (Sinh[γ ϵ ℏ/2] (t (2 a + γ) - 2 a (a + γ) ϵ + 2 ϖ) ℏ / (2 γ)));
```

```
{S𝔻$P = (Cosh[ℏ ((ϵ - t)/2 + ϵ a)] - Cosh[ℏ √((t^2 + ϵ^2)/4 + ϵ ϖ)]) /
    (ℏ Sinh[-ϵ ℏ/2] (ϖ - ϵ a^2 + (t - ϵ) a + t/2)),
 Simplify[S𝔻$P == (S𝔻$P /. {a → -a - 1, t → -t})] // HL,
 PowerExpand@Simplify[(S𝔻$P /. {ℏ → γ^2 ℏ, ϵ → ϵ/γ, a → a/γ, t → γ^-2 t, ϖ → γ^-3 ϖ}) ==
    S𝔻$g (S𝔻$g /. {a → -a - γ, t → -t})] // HL,
 S𝔻$Q = Simplify[S𝔻$P /. {a → c - 1/2}],
 Simplify[S𝔻$Q == (S𝔻$Q /. {c → -c, t → -t})] // HL,
 Simplify[S𝔻$g == FullSimplify[
    √S𝔻$Q /. c → a + 1/2 /. {ℏ → γ^2 ℏ, ϵ → ϵ/γ, a → a/γ, t → γ^-2 t, ϖ → γ^-3 ϖ}]] // HL
}
```

$$\left\{-\left(\left(\left(\mathrm{Cosh}\left[\left(a\,\epsilon+\frac{1}{2}(-t+\epsilon)\right)\hbar\right]-\mathrm{Cosh}\left[\sqrt{\frac{1}{4}(t^2+\epsilon^2)+\epsilon\,\varpi}\;\hbar\right]\right)\mathrm{Csch}\left[\frac{\epsilon\,\hbar}{2}\right]\right)\Big/\left(\left(\frac{t}{2}+a\,(t-\epsilon)-a^2\epsilon+\varpi\right)\hbar\right)\right),\ \mathrm{True},\ \mathrm{True},\right.$$

$$\left.-\left(\left(4\left(\mathrm{Cosh}\left[\frac{1}{2}(t-2\,c\,\epsilon)\,\hbar\right]-\mathrm{Cosh}\left[\frac{1}{2}\sqrt{t^2+\epsilon^2+4\,\epsilon\,\varpi}\;\hbar\right]\right)\mathrm{Csch}\left[\frac{\epsilon\,\hbar}{2}\right]\right)\Big/\left(\left(4\,c\,t+\epsilon-4\,c^2\epsilon+4\,\varpi\right)\hbar\right)\right),\ \mathrm{True},\ \mathrm{True}\right\}$$

```
FullSimplify[S𝔻$g /. {ℏ → γ^2 ℏ, ϵ → ϵ/γ, a → a/γ, t → γ^-2 t, ϖ → γ^-3 ϖ}]
```

$$\sqrt{2}\sqrt{\left(\left(\gamma\left(-\mathrm{Cosh}\left[\frac{1}{2}\left(t-(2\,a+\gamma)\,\epsilon\right)\hbar\right]+\mathrm{Cosh}\left[\frac{1}{2}\sqrt{t^2+\gamma^2\epsilon^2+4\,\epsilon\,\varpi}\;\hbar\right]\right)\mathrm{Csch}\left[\frac{\gamma\,\epsilon\,\hbar}{2}\right]\right)\Big/\left(\left(t\,(2\,a+\gamma)-2\,a\,(a+\gamma)\,\epsilon+2\,\varpi\right)\hbar\right)\right)}$$

```
SS[S𝔻$g]
```

$$1+\left(\frac{t^2}{48}+\left(\frac{a}{24}+\frac{a^2}{24}\right)\epsilon^2+\epsilon\left(-\frac{t}{48}-\frac{a\,t}{24}+\frac{\varpi}{24}\right)\right)\hbar^2+$$
$$\left(\frac{t^4}{23040}+\epsilon\left(-\frac{t^3}{11520}-\frac{a\,t^3}{5760}+\frac{t^2\varpi}{5760}\right)+\epsilon^2\left(-\frac{t^2}{4608}+\frac{a\,t^2}{1440}+\frac{a^2t^2}{1440}+\frac{t\,\varpi}{5760}+\frac{a\,t\,\varpi}{2880}+\frac{\varpi^2}{1920}\right)\right)\hbar^4$$

SDeq

```
S𝔻$f = FullSimplify[ⅇ^(ℏ (t/2-ϵ a)) (S𝔻$g /. {a → -a, t → -t})];
```

```
SS[S𝔻$f]
```

$$1+\left(\frac{t}{2}-a\,\epsilon\right)\hbar+\left(\frac{7t^2}{48}+\left(-\frac{a}{24}+\frac{13a^2}{24}\right)\epsilon^2+\epsilon\left(\frac{t}{48}-\frac{13\,a\,t}{24}+\frac{\varpi}{24}\right)\right)\hbar^2+\left(\frac{t^3}{32}+\epsilon^2\left(-\frac{a\,t}{24}+\frac{5a^2t}{16}-\frac{a\,\varpi}{24}\right)+\epsilon\left(\frac{t^2}{96}-\frac{a\,t^2}{6}+\frac{t\,\varpi}{48}\right)\right)\hbar^3+\left(\frac{121\,t^4}{23040}+\epsilon\left(\frac{31\,t^3}{11520}-\frac{211\,a\,t^3}{5760}+\frac{31\,t^2\varpi}{5760}\right)+\epsilon^2\left(-\frac{t^2}{4608}-\frac{47\,a\,t^2}{2880}+\frac{287\,a^2t^2}{2880}-\frac{t\,\varpi}{5760}-\frac{59\,a\,t\,\varpi}{2880}+\frac{\varpi^2}{1920}\right)\right)\hbar^4+\left(\frac{11\,t^5}{15360}+\epsilon\left(\frac{11\,t^4}{23040}-\frac{143\,a\,t^4}{23040}+\frac{11\,t^3\varpi}{11520}\right)+\epsilon^2\left(-\frac{t^3}{9216}-\frac{a\,t^3}{256}+\frac{a^2t^3}{45}-\frac{t^2\varpi}{11520}-\frac{1}{192}a\,t^2\varpi+\frac{t\,\varpi^2}{3840}\right)\right)\hbar^5$$

SDeq

```
S𝔻$ϖ = γ CU[y, x] + ϵ CU[a, a] - (t - γ ϵ) CU[a] - t γ CU[] / 2;
```

SDeq

```
DeclareMorphism[S𝔻, QU → CU, {a → CU@a,
  x → 𝕊CU[SS[S𝔻$f], a → CU[a], ϖ → S𝔻$ϖ] ** CU@x,
  y → 𝕊CU[SS[S𝔻$g] , a → CU[a], ϖ → S𝔻$ϖ] ** CU@y
 }]
```

Verifying the θ-symmetry:

```
Table[HL@Simplify[Cθ[S𝔻[z]] == S𝔻[Qθ[z]]], {z, QU /@ {y, a, x}}]
```

```
{True, True, True}
```

Verifying that the symmetric dequantizator is a homomorphism:

```
With[{bas = QU /@ {y, a, x}},
 Table[{z1, z2} → HL@Simp[S𝔻[z1 ** z2] - S𝔻[z1] ** S𝔻[z2]], {z1, bas}, {z2, bas}] ]
```

```
{{{QU[y], QU[y]} → 0, {QU[y], QU[a]} → 0, {QU[y], QU[x]} → 0},
 {{QU[a], QU[y]} → 0, {QU[a], QU[a]} → 0, {QU[a], QU[x]} → 0},
 {{QU[x], QU[y]} → 0, {QU[x], QU[a]} → 0, {QU[x], QU[x]} → 0}}
```